Rules for integrands of the form $(a + b \,\mathrm{Sinh}[c + d\, x^n])^p$

1. $\int (a + b \,\mathrm{Sinh}[c + d\, x^n])^p \, dx$ when $n \in \mathbb{Z} \land p \in \mathbb{Z}$

   1. $\int (a + b \,\mathrm{Sinh}[c + d\, x^n])^p \, dx$ when $n - 1 \in \mathbb{Z}^+ \land p \in \mathbb{Z}^+$

      1: $\int \mathrm{Sinh}[c + d\, x^n] \, dx$ when $n - 1 \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: $\mathrm{Sinh}[z] == \frac{e^z}{2} - \frac{e^{-z}}{2}$

Basis: $\mathrm{Cosh}[z] == \frac{e^z}{2} + \frac{e^{-z}}{2}$

Rule: If $n - 1 \in \mathbb{Z}^+$, then

$$\int \mathrm{Sinh}[c + d\, x^n] \, dx \longrightarrow \frac{1}{2} \int e^{c + d\, x^n} \, dx - \frac{1}{2} \int e^{-c - d\, x^n} \, dx$$

Program code:

```
Int[Sinh[c_.+d_.*x_^n_],x_Symbol] :=
  1/2*Int[E^(c+d*x^n),x] - 1/2*Int[E^(-c-d*x^n),x] /;
FreeQ[{c,d},x] && IGtQ[n,1]
```

```
Int[Cosh[c_.+d_.*x_^n_],x_Symbol] :=
  1/2*Int[E^(c+d*x^n),x] + 1/2*Int[E^(-c-d*x^n),x] /;
FreeQ[{c,d},x] && IGtQ[n,1]
```

2: $\int (a + b \operatorname{Sinh}[c + d x^n])^p \, dx$ when $n - 1 \in \mathbb{Z}^+ \land p - 1 \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $n - 1 \in \mathbb{Z}^+ \land p - 1 \in \mathbb{Z}^+$, then

$$\int (a + b \operatorname{Sinh}[c + d x^n])^p \, dx \to \int \operatorname{TrigReduce}[(a + b \operatorname{Sinh}[c + d x^n])^p, x] \, dx$$

Program code:

```
Int[(a_.+b_.*Sinh[c_.+d_.*x_^n_])^p_,x_Symbol] :=
  Int[ExpandTrigReduce[(a+b*Sinh[c+d*x^n])^p,x],x] /;
FreeQ[{a,b,c,d},x] && IGtQ[n,1] && IGtQ[p,1]
```

```
Int[(a_.+b_.*Cosh[c_.+d_.*x_^n_])^p_,x_Symbol] :=
  Int[ExpandTrigReduce[(a+b*Cosh[c+d*x^n])^p,x],x] /;
FreeQ[{a,b,c,d},x] && IGtQ[n,1] && IGtQ[p,1]
```

2: $\int (a + b \operatorname{Sinh}[c + d x^n])^p \, dx$ when $n \in \mathbb{Z}^- \land p \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: If $n \in \mathbb{Z}$, then $F[x^n] = -\operatorname{Subst}\left[\frac{F[x^{-n}]}{x^2}, x, \frac{1}{x}\right] \partial_x \frac{1}{x}$

Rule: If $n \in \mathbb{Z}^- \land p \in \mathbb{Z}$, then

$$\int (a + b \operatorname{Sinh}[c + d x^n])^p \, dx \longrightarrow -\operatorname{Subst}\left[\int \frac{(a + b \operatorname{Sinh}[c + d x^{-n}])^p}{x^2} \, dx, x, \frac{1}{x}\right]$$

Program code:

```
Int[(a_.+b_.*Sinh[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  -Subst[Int[(a+b*Sinh[c+d*x^(-n)])^p/x^2,x],x,1/x] /;
FreeQ[{a,b,c,d},x] && ILtQ[n,0] && IntegerQ[p]
```

```
Int[(a_.+b_.*Cosh[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  -Subst[Int[(a+b*Cosh[c+d*x^(-n)])^p/x^2,x],x,1/x] /;
FreeQ[{a,b,c,d},x] && ILtQ[n,0] && IntegerQ[p]
```

2: $\int (a + b \operatorname{Sinh}[c + d x^n])^p \, dx$ when $n \in \mathbb{F} \land p \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: If $k \in \mathbb{Z}^+$, then $F[x^n] == k \operatorname{Subst}[x^{k-1} F[x^{k n}], x, x^{1/k}] \, \partial_x x^{1/k}$

Rule: If $n \in \mathbb{F} \land p \in \mathbb{Z}$, let $k = \operatorname{Denominator}[n]$, then

$$\int (a + b \operatorname{Sinh}[c + d x^n])^p \, dx \longrightarrow k \operatorname{Subst}\left[\int x^{k-1} (a + b \operatorname{Sinh}[c + d x^{k n}])^p \, dx, x, x^{1/k}\right]$$

Program code:

```
Int[(a_.+b_.*Sinh[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  Module[{k=Denominator[n]},
  k*Subst[Int[x^(k-1)*(a+b*Sinh[c+d*x^(k*n)])^p,x],x,x^(1/k)]] /;
FreeQ[{a,b,c,d},x] && FractionQ[n] && IntegerQ[p]
```

```
Int[(a_.+b_.*Cosh[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  Module[{k=Denominator[n]},
  k*Subst[Int[x^(k-1)*(a+b*Cosh[c+d*x^(k*n)])^p,x],x,x^(1/k)]] /;
FreeQ[{a,b,c,d},x] && FractionQ[n] && IntegerQ[p]
```

3. $\int (a + b \,\text{Sinh}[c + d\, x^n])^p \, dx$ when $p \in \mathbb{Z}^+$

1: $\int \text{Sinh}[c + d\, x^n] \, dx$

Derivation: Algebraic expansion

Basis: $\text{Sinh}[z] == \frac{e^z}{2} - \frac{e^{-z}}{2}$

Basis: $\text{Cosh}[z] == \frac{e^z}{2} + \frac{e^{-z}}{2}$

Rule:

$$\int \text{Sinh}[c + d\, x^n] \, dx \longrightarrow \frac{1}{2} \int e^{c+d\, x^n} \, dx - \frac{1}{2} \int e^{-c-d\, x^n} \, dx$$

Program code:

```
Int[Sinh[c_.+d_.*x_^n_],x_Symbol] :=
  1/2*Int[E^(c+d*x^n),x] - 1/2*Int[E^(-c-d*x^n),x] /;
FreeQ[{c,d,n},x]
```

```
Int[Cosh[c_.+d_.*x_^n_],x_Symbol] :=
  1/2*Int[E^(c+d*x^n),x] + 1/2*Int[E^(-c-d*x^n),x] /;
FreeQ[{c,d,n},x]
```

2: $\int (a + b \operatorname{Sinh}[c + d x^n])^p \, dx$ when $p \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $p \in \mathbb{Z}^+$, then

$$\int (a + b \operatorname{Sinh}[c + d x^n])^p \, dx \longrightarrow \int \operatorname{TrigReduce}[(a + b \operatorname{Sinh}[c + d x^n])^p, x] \, dx$$

Program code:

```
Int[(a_.+b_.*Sinh[c_.+d_.*x_^n_])^p_,x_Symbol] :=
  Int[ExpandTrigReduce[(a+b*Sinh[c+d*x^n])^p,x],x] /;
FreeQ[{a,b,c,d,n},x] && IGtQ[p,0]
```

```
Int[(a_.+b_.*Cosh[c_.+d_.*x_^n_])^p_,x_Symbol] :=
  Int[ExpandTrigReduce[(a+b*Cosh[c+d*x^n])^p,x],x] /;
FreeQ[{a,b,c,d,n},x] && IGtQ[p,0]
```

S: $\int (a + b \operatorname{Sinh}[c + d u^n])^p \, dx$ when $p \in \mathbb{Z} \land u == e + f x$

- Derivation: Integration by substitution
- Rule: If $p \in \mathbb{Z} \land u == e + f x$, then

$$\int (a + b \operatorname{Sinh}[c + d u^n])^p \, dx \longrightarrow \frac{1}{f} \operatorname{Subst}\left[\int (a + b \operatorname{Sinh}[c + d x^n])^p \, dx, x, u\right]$$

- Program code:

```
Int[(a_.+b_.*Sinh[c_.+d_.*u_^n_])^p_.,x_Symbol] :=
  1/Coefficient[u,x,1]*Subst[Int[(a+b*Sinh[c+d*x^n])^p,x],x,u] /;
FreeQ[{a,b,c,d,n},x] && IntegerQ[p] && LinearQ[u,x] && NeQ[u,x]
```

```
Int[(a_.+b_.*Cosh[c_.+d_.*u_^n_])^p_.,x_Symbol] :=
  1/Coefficient[u,x,1]*Subst[Int[(a+b*Cosh[c+d*x^n])^p,x],x,u] /;
FreeQ[{a,b,c,d,n},x] && IntegerQ[p] && LinearQ[u,x] && NeQ[u,x]
```

X: $\int (a + b \operatorname{Sinh}[c + d u^n])^p \, dx$

- Rule:

$$\int (a + b \operatorname{Sinh}[c + d u^n])^p \, dx \longrightarrow \int (a + b \operatorname{Sinh}[c + d u^n])^p \, dx$$

- Program code:

```
Int[(a_.+b_.*Sinh[c_.+d_.*u_^n_])^p_,x_Symbol] :=
  Unintegrable[(a+b*Sinh[c+d*u^n])^p,x] /;
FreeQ[{a,b,c,d,n,p},x] && LinearQ[u,x]
```

```
Int[(a_.+b_.*Cosh[c_.+d_.*u_^n_])^p_,x_Symbol] :=
  Unintegrable[(a+b*Cosh[c+d*u^n])^p,x] /;
FreeQ[{a,b,c,d,n,p},x] && LinearQ[u,x]
```

N: $\int (a + b\,\text{Sinh}[u])^p \, dx$ when $u == c + d\,x^n$

Derivation: Algebraic normalization

– Rule: If $u == c + d\,x^n$, then

$$\int (a + b\,\text{Sinh}[u])^p \, dx \longrightarrow \int \left(a + b\,\text{Sinh}[c + d\,x^n]\right)^p dx$$

Program code:

```
Int[(a_.+b_.*Sinh[u_])^p_.,x_Symbol] :=
  Int[(a+b*Sinh[ExpandToSum[u,x]])^p,x] /;
FreeQ[{a,b,p},x] && BinomialQ[u,x] && Not[BinomialMatchQ[u,x]]
```

```
Int[(a_.+b_.*Cosh[u_])^p_.,x_Symbol] :=
  Int[(a+b*Cosh[ExpandToSum[u,x]])^p,x] /;
FreeQ[{a,b,p},x] && BinomialQ[u,x] && Not[BinomialMatchQ[u,x]]
```

Rules for integrands of the form $(e\,x)^m \,(a + b\,\text{Sinh}[c + d\,x^n])^p$

1. $\int (e\,x)^m \left(a + b\,\text{Sinh}[c + d\,x^n]\right)^p dx$ when $\frac{m+1}{n} \in \mathbb{Z}$

   1. $\int x^m \left(a + b\,\text{Sinh}[c + d\,x^n]\right)^p dx$ when $\frac{m+1}{n} \in \mathbb{Z}$

      1. $\int \frac{\text{Sinh}[c + d\,x^n]}{x} \, dx$

1: $\int \frac{\text{Sinh}[d\, x^n]}{x}\, dx$

Derivation: Primitive rule

Basis: $\text{SinhIntegral}'[z] == \frac{\text{Sinh}[z]}{z}$

–

Rule:

$$\int \frac{\text{Sinh}[d\, x^n]}{x}\, dx \longrightarrow \frac{\text{SinhIntegral}[d\, x^n]}{n}$$

–

Program code:

```
Int[Sinh[d_.*x_^n_]/x_,x_Symbol] :=
  SinhIntegral[d*x^n]/n /;
FreeQ[{d,n},x]
```

```
Int[Cosh[d_.*x_^n_]/x_,x_Symbol] :=
  CoshIntegral[d*x^n]/n /;
FreeQ[{d,n},x]
```

2: $\int \frac{\text{Sinh}[c + d x^n]}{x} \, dx$

Derivation: Algebraic expansion

Basis: $\text{Sinh}[w + z] == \text{Sinh}[w] \, \text{Cosh}[z] + \text{Cosh}[w] \, \text{Sinh}[z]$

–

Rule:

$$\int \frac{\text{Sinh}[c + d x^n]}{x} \, dx \longrightarrow \text{Sinh}[c] \int \frac{\text{Cosh}[d x^n]}{x} \, dx + \text{Cosh}[c] \int \frac{\text{Sinh}[d x^n]}{x} \, dx$$

–

Program code:

```
Int[Sinh[c_+d_.*x_^n_]/x_,x_Symbol] :=
  Sinh[c]*Int[Cosh[d*x^n]/x,x] + Cosh[c]*Int[Sinh[d*x^n]/x,x] /;
FreeQ[{c,d,n},x]
```

```
Int[Cosh[c_+d_.*x_^n_]/x_,x_Symbol] :=
  Cosh[c]*Int[Cosh[d*x^n]/x,x] + Sinh[c]*Int[Sinh[d*x^n]/x,x] /;
FreeQ[{c,d,n},x]
```

2: $\int x^m \left(a + b \operatorname{Sinh}[c + d x^n]\right)^p dx$ when $\frac{m+1}{n} \in \mathbb{Z} \land \left(p == 1 \lor m == n - 1 \lor p \in \mathbb{Z} \land \frac{m+1}{n} > 0\right)$

Derivation: Integration by substitution

Basis: If $\frac{m+1}{n} \in \mathbb{Z}$, then $x^m F[x^n] == \frac{1}{n} \operatorname{Subst}\left[x^{\frac{m+1}{n}-1} F[x], x, x^n\right] \partial_x x^n$

Rule: If $\frac{m+1}{n} \in \mathbb{Z} \land \left(p == 1 \lor m == n - 1 \lor p \in \mathbb{Z} \land \frac{m+1}{n} > 0\right)$, then

$$\int x^m \left(a + b \operatorname{Sinh}[c + d x^n]\right)^p dx \longrightarrow \frac{1}{n} \operatorname{Subst}\left[\int x^{\frac{m+1}{n}-1} \left(a + b \operatorname{Sinh}[c + d x]\right)^p dx, x, x^n\right]$$

Program code:

```
Int[x_^m_.*(a_.+b_.*Sinh[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  1/n*Subst[Int[x^(Simplify[(m+1)/n]-1)*(a+b*Sinh[c+d*x])^p,x],x,x^n] /;
FreeQ[{a,b,c,d,m,n,p},x] && IntegerQ[Simplify[(m+1)/n]] && (EqQ[p,1] || EqQ[m,n-1] || IntegerQ[p] && GtQ[Simplify[(m+1)/n],0])
```

```
Int[x_^m_.*(a_.+b_.*Cosh[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  1/n*Subst[Int[x^(Simplify[(m+1)/n]-1)*(a+b*Cosh[c+d*x])^p,x],x,x^n] /;
FreeQ[{a,b,c,d,m,n,p},x] && IntegerQ[Simplify[(m+1)/n]] && (EqQ[p,1] || EqQ[m,n-1] || IntegerQ[p] && GtQ[Simplify[(m+1)/n],0])
```

2: $\int (e\,x)^m\,(a+b\,\mathrm{Sinh}[c+d\,x^n])^p\,dx$ when $\frac{m+1}{n}\in\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(e\,x)^m}{x^m} == 0$

Rule: If $\frac{m+1}{n}\in\mathbb{Z}$, then

$$\int (e\,x)^m\,(a+b\,\mathrm{Sinh}[c+d\,x^n])^p\,dx \longrightarrow \frac{e^{\mathrm{IntPart}[m]}\,(e\,x)^{\mathrm{FracPart}[m]}}{x^{\mathrm{FracPart}[m]}}\int x^m\,(a+b\,\mathrm{Sinh}[c+d\,x^n])^p\,dx$$

Program code:

```
Int[(e_*x_)^m_*(a_.+b_.*Sinh[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  e^IntPart[m]*(e*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a+b*Sinh[c+d*x^n])^p,x] /;
FreeQ[{a,b,c,d,e,m,n,p},x] && IntegerQ[Simplify[(m+1)/n]]
```

```
Int[(e_*x_)^m_*(a_.+b_.*Cosh[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  e^IntPart[m]*(e*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a+b*Cosh[c+d*x^n])^p,x] /;
FreeQ[{a,b,c,d,e,m,n,p},x] && IntegerQ[Simplify[(m+1)/n]]
```

2. $\int (e\,x)^m\,(a+b\,\text{Sinh}[c+d\,x^n])^p\,dx$ when $p\in\mathbb{Z}\land n\in\mathbb{Z}$

1. $\int (e\,x)^m\,(a+b\,\text{Sinh}[c+d\,x^n])^p\,dx$ when $p\in\mathbb{Z}\land n\in\mathbb{Z}^+$

1. $\int (e\,x)^m\,\text{Sinh}[c+d\,x^n]\,dx$

1: $\int (e\,x)^m\,\text{Sinh}[c+d\,x^n]\,dx$ when $n\in\mathbb{Z}^+\land 0<n<m+1$

Reference: CRC 392, A&S 4.3.119

Reference: CRC 396, A&S 4.3.123

Derivation: Integration by parts

Basis: If $n\in\mathbb{Z}$, then $(e\,x)^m\,\text{Sinh}[c+d\,x^n]=-\frac{e^{n-1}\,(e\,x)^{m-n+1}}{d\,n}\,\partial_x\text{Cosh}[c+d\,x^n]$

Rule: If $n\in\mathbb{Z}^+\land 0<n<m+1$, then

$$\int (e\,x)^m\,\text{Sinh}[c+d\,x^n]\,dx\;\longrightarrow\;\frac{e^{n-1}\,(e\,x)^{m-n+1}\,\text{Cosh}[c+d\,x^n]}{d\,n}-\frac{e^n\,(m-n+1)}{d\,n}\int (e\,x)^{m-n}\,\text{Cosh}[c+d\,x^n]\,dx$$

Program code:

```
Int[(e_.*x_)^m_.*Sinh[c_.+d_.*x_^n_],x_Symbol] :=
  e^(n-1)*(e*x)^(m-n+1)*Cosh[c+d*x^n]/(d*n) -
   e^n*(m-n+1)/(d*n)*Int[(e*x)^(m-n)*Cosh[c+d*x^n],x] /;
FreeQ[{c,d,e},x] && IGtQ[n,0] && LtQ[0,n,m+1]
```

```
Int[(e_.*x_)^m_.*Cosh[c_.+d_.*x_^n_],x_Symbol] :=
  e^(n-1)*(e*x)^(m-n+1)*Sinh[c+d*x^n]/(d*n) -
   e^n*(m-n+1)/(d*n)*Int[(e*x)^(m-n)*Sinh[c+d*x^n],x] /;
FreeQ[{c,d,e},x] && IGtQ[n,0] && LtQ[0,n,m+1]
```

2: $\int (e\,x)^m \operatorname{Sinh}\left[c+d\,x^n\right] dx$ when $n \in \mathbb{Z}^+ \land m < -1$

Reference: CRC 405, A&S 4.3.120

Reference: CRC 406, A&S 4.3.124

Derivation: Integration by parts

Rule: If $n \in \mathbb{Z}^+ \land m < -1$, then

$$\int (e\,x)^m \operatorname{Sinh}\left[c+d\,x^n\right] dx \longrightarrow \frac{(e\,x)^{m+1} \operatorname{Sinh}\left[c+d\,x^n\right]}{e\,(m+1)} - \frac{d\,n}{e^n\,(m+1)} \int (e\,x)^{m+n} \operatorname{Cosh}\left[c+d\,x^n\right] dx$$

Program code:

```
Int[(e_.*x_)^m_*Sinh[c_.+d_.*x_^n_],x_Symbol] :=
  (e*x)^(m+1)*Sinh[c+d*x^n]/(e*(m+1)) -
  d*n/(e^n*(m+1))*Int[(e*x)^(m+n)*Cosh[c+d*x^n],x] /;
FreeQ[{c,d,e},x] && IGtQ[n,0] && LtQ[m,-1]
```

```
Int[(e_.*x_)^m_*Cosh[c_.+d_.*x_^n_],x_Symbol] :=
  (e*x)^(m+1)*Cosh[c+d*x^n]/(e*(m+1)) -
  d*n/(e^n*(m+1))*Int[(e*x)^(m+n)*Sinh[c+d*x^n],x] /;
FreeQ[{c,d,e},x] && IGtQ[n,0] && LtQ[m,-1]
```

3: $\int (e\,x)^m \,\text{Sinh}\left[c+d\,x^n\right] dx$ when $n \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: $\text{Sinh}[z] == \frac{e^z}{2} - \frac{e^{-z}}{2}$

Basis: $\text{Cosh}[z] == \frac{e^z}{2} + \frac{e^{-z}}{2}$

Rule: If $n \in \mathbb{Z}^+$, then

$$\int (e\,x)^m \,\text{Sinh}\left[c+d\,x^n\right] dx \rightarrow \frac{1}{2}\int (e\,x)^m\, e^{c+d\,x^n}\, dx - \frac{1}{2}\int (e\,x)^m\, e^{-c-d\,x^n}\, dx$$

Program code:

```
Int[(e_.*x_)^m_.*Sinh[c_.+d_.*x_^n_],x_Symbol] :=
  1/2*Int[(e*x)^m*E^(c+d*x^n),x] - 1/2*Int[(e*x)^m*E^(-c-d*x^n),x] /;
FreeQ[{c,d,e,m},x] && IGtQ[n,0]
```

```
Int[(e_.*x_)^m_.*Cosh[c_.+d_.*x_^n_],x_Symbol] :=
  1/2*Int[(e*x)^m*E^(c+d*x^n),x] + 1/2*Int[(e*x)^m*E^(-c-d*x^n),x] /;
FreeQ[{c,d,e,m},x] && IGtQ[n,0]
```

2. $\int (e\,x)^m \left(a+b\,\text{Sinh}\left[c+d\,x^n\right]\right)^p dx$ when $p>1$

1: $\int \frac{\text{Sinh}\left[a+b\,x^n\right]^p}{x^n} dx$ when $(n \mid p) \in \mathbb{Z} \land p>1 \land n \neq 1$

Derivation: Integration by parts

Rule: If $(n \mid p) \in \mathbb{Z} \land p>1 \land n \neq 1$, then

$$\int \frac{\text{Sinh}\left[a+b\,x^n\right]^p}{x^n}\,dx \longrightarrow -\frac{\text{Sinh}\left[a+b\,x^n\right]^p}{(n-1)\,x^{n-1}}+\frac{b\,n\,p}{n-1}\int \text{Sinh}\left[a+b\,x^n\right]^{p-1}\,\text{Cosh}\left[a+b\,x^n\right]\,dx$$

Program code:

```
Int[x_^m_.*Sinh[a_.+b_.*x_^n_]^p_,x_Symbol] :=
  -Sinh[a+b*x^n]^p/((n-1)*x^(n-1)) +
  b*n*p/(n-1)*Int[Sinh[a+b*x^n]^(p-1)*Cosh[a+b*x^n],x] /;
FreeQ[{a,b},x] && IntegersQ[n,p] && EqQ[m+n,0] && GtQ[p,1] && NeQ[n,1]
```

```
Int[x_^m_.*Cosh[a_.+b_.*x_^n_]^p_,x_Symbol] :=
  -Cosh[a+b*x^n]^p/((n-1)*x^(n-1)) +
  b*n*p/(n-1)*Int[Cosh[a+b*x^n]^(p-1)*Sinh[a+b*x^n],x] /;
FreeQ[{a,b},x] && IntegersQ[n,p] && EqQ[m+n,0] && GtQ[p,1] && NeQ[n,1]
```

2: $\int x^m\,\text{Sinh}\left[a+b\,x^n\right]^p\,dx$ when $m-2\,n+1=0\,\land\,p>1$

Reference: G&R 2.471.1b' special case when $m-2\,n+1==0$

Reference: G&R 2.471.1a' special case with $m-2\,n+1==0$

Rule: If $m-2\,n+1==0\,\land\,p>1$, then

$$\int x^m\,\text{Sinh}\left[a+b\,x^n\right]^p\,dx \longrightarrow -\frac{n\,\text{Sinh}\left[a+b\,x^n\right]^p}{b^2\,n^2\,p^2}+\frac{x^n\,\text{Cosh}\left[a+b\,x^n\right]\,\text{Sinh}\left[a+b\,x^n\right]^{p-1}}{b\,n\,p}-\frac{p-1}{p}\int x^m\,\text{Sinh}\left[a+b\,x^n\right]^{p-2}\,dx$$

–

Program code:

```
Int[x_^m_.*Sinh[a_.+b_.*x_^n_]^p_,x_Symbol] :=
  -n*Sinh[a+b*x^n]^p/(b^2*n^2*p^2) +
  x^n*Cosh[a+b*x^n]*Sinh[a+b*x^n]^(p-1)/(b*n*p) -
  (p-1)/p*Int[x^m*Sinh[a+b*x^n]^(p-2),x] /;
FreeQ[{a,b,m,n},x] && EqQ[m-2*n+1] && GtQ[p,1]
```

```
Int[x_^m_.*Cosh[a_.+b_.*x_^n_]^p_,x_Symbol] :=
  -n*Cosh[a+b*x^n]^p/(b^2*n^2*p^2) +
  x^n*Sinh[a+b*x^n]*Cosh[a+b*x^n]^(p-1)/(b*n*p) +
  (p-1)/p*Int[x^m*Cosh[a+b*x^n]^(p-2),x] /;
FreeQ[{a,b,m,n},x] && EqQ[m-2*n+1] && GtQ[p,1]
```

3: $\int x^m \operatorname{Sinh}[a+b x^n]^p \, dx$ when $(m \mid n) \in \mathbb{Z} \land p > 1 \land 0 < 2n < m+1$

Reference: G&R 2.471.1b'

Reference: G&R 2.631.3'

Rule: If $(m \mid n) \in \mathbb{Z} \land p > 1 \land 0 < 2n < m+1$, then

$$\int x^m \operatorname{Sinh}[a+b x^n]^p \, dx \longrightarrow$$

$$-\frac{(m-n+1)\, x^{m-2n+1} \operatorname{Sinh}[a+b x^n]^p}{b^2 n^2 p^2} + \frac{x^{m-n+1} \operatorname{Cosh}[a+b x^n] \operatorname{Sinh}[a+b x^n]^{p-1}}{b n p} - \frac{p-1}{p} \int x^m \operatorname{Sinh}[a+b x^n]^{p-2} \, dx + \frac{(m-n+1)\,(m-2n+1)}{b^2 n^2 p^2} \int x^{m-2n} \operatorname{Sinh}[a+b x^n]^p \, dx$$

Program code:

```
Int[x_^m_.*Sinh[a_.+b_.*x_^n_]^p_,x_Symbol] :=
  -(m-n+1)*x^(m-2*n+1)*Sinh[a+b*x^n]^p/(b^2*n^2*p^2) +
  x^(m-n+1)*Cosh[a+b*x^n]*Sinh[a+b*x^n]^(p-1)/(b*n*p) -
  (p-1)/p*Int[x^m*Sinh[a+b*x^n]^(p-2),x] +
  (m-n+1)*(m-2*n+1)/(b^2*n^2*p^2)*Int[x^(m-2*n)*Sinh[a+b*x^n]^p,x] /;
FreeQ[{a,b},x] && IntegersQ[m,n] && GtQ[p,1] && LtQ[0,2*n,m+1]
```

```
Int[x_^m_.*Cosh[a_.+b_.*x_^n_]^p_,x_Symbol] :=
  -(m-n+1)*x^(m-2*n+1)*Cosh[a+b*x^n]^p/(b^2*n^2*p^2) +
  x^(m-n+1)*Sinh[a+b*x^n]*Cosh[a+b*x^n]^(p-1)/(b*n*p) +
  (p-1)/p*Int[x^m*Cosh[a+b*x^n]^(p-2),x] +
  (m-n+1)*(m-2*n+1)/(b^2*n^2*p^2)*Int[x^(m-2*n)*Cosh[a+b*x^n]^p,x] /;
FreeQ[{a,b},x] && IntegersQ[m,n] && GtQ[p,1] && LtQ[0,2*n,m+1]
```

4: $\int x^m \operatorname{Sinh}\left[a+b x^n\right]^p \, dx$ when $(m \mid n) \in \mathbb{Z} \land p>1 \land 0<2n<1-m \land m+n+1 \neq 0$

Reference: G&R 2.475.1'

Reference: G&R 2.475.2'

Rule: If $(m \mid n) \in \mathbb{Z} \land p>1 \land 0<2n<1-m \land m+n+1 \neq 0$, then

$$\int x^m \operatorname{Sinh}\left[a+b x^n\right]^p \, dx \longrightarrow$$

$$\frac{x^{m+1} \operatorname{Sinh}\left[a+b x^n\right]^p}{m+1}-\frac{b\, n\, p\, x^{m+n+1} \operatorname{Cosh}\left[a+b x^n\right] \operatorname{Sinh}\left[a+b x^n\right]^{p-1}}{(m+1)\,(m+n+1)}+\frac{b^2 n^2 p^2}{(m+1)\,(m+n+1)} \int x^{m+2n} \operatorname{Sinh}\left[a+b x^n\right]^p \, dx+\frac{b^2 n^2 p\,(p-1)}{(m+1)\,(m+n+1)} \int x^{m+2n} \operatorname{Sinh}\left[a+b x^n\right]^{p-2} \, dx$$

–

Program code:

```
Int[x_^m_.*Sinh[a_.+b_.*x_^n_]^p_,x_Symbol] :=
  x^(m+1)*Sinh[a+b*x^n]^p/(m+1) -
  b*n*p*x^(m+n+1)*Cosh[a+b*x^n]*Sinh[a+b*x^n]^(p-1)/((m+1)*(m+n+1)) +
  b^2*n^2*p^2/((m+1)*(m+n+1))*Int[x^(m+2*n)*Sinh[a+b*x^n]^p,x] +
  b^2*n^2*p*(p-1)/((m+1)*(m+n+1))*Int[x^(m+2*n)*Sinh[a+b*x^n]^(p-2),x] /;
FreeQ[{a,b},x] && IntegersQ[m,n] && GtQ[p,1] && LtQ[0,2*n,1-m] && NeQ[m+n+1,0]
```

```
Int[x_^m_.*Cosh[a_.+b_.*x_^n_]^p_,x_Symbol] :=
  x^(m+1)*Cosh[a+b*x^n]^p/(m+1) -
  b*n*p*x^(m+n+1)*Sinh[a+b*x^n]*Cosh[a+b*x^n]^(p-1)/((m+1)*(m+n+1)) +
  b^2*n^2*p^2/((m+1)*(m+n+1))*Int[x^(m+2*n)*Cosh[a+b*x^n]^p,x] -
  b^2*n^2*p*(p-1)/((m+1)*(m+n+1))*Int[x^(m+2*n)*Cosh[a+b*x^n]^(p-2),x] /;
FreeQ[{a,b},x] && IntegersQ[m,n] && GtQ[p,1] && LtQ[0,2*n,1-m] && NeQ[m+n+1,0]
```

5: $\int (e\,x)^m\,(a+b\,\mathrm{Sinh}[c+d\,x^n])^p\,dx$ when $p\in\mathbb{Z}\,\land\,n\in\mathbb{Z}^+\,\land\,m\in\mathbb{F}$

Derivation: Integration by substitution

Basis: If $k\in\mathbb{Z}^+$, then $(e\,x)^m\,F[x] = \frac{k}{e}\,\mathrm{Subst}\left[x^{k\,(m+1)-1}\,F\left[\frac{x^k}{e}\right],\,x,\,(e\,x)^{1/k}\right]\,\partial_x\,(e\,x)^{1/k}$

Rule: If $p\in\mathbb{Z}\,\land\,n\in\mathbb{Z}^+\,\land\,m\in\mathbb{F}$, let $k=\mathrm{Denominator}[m]$, then

$$\int (e\,x)^m\,(a+b\,\mathrm{Sinh}[c+d\,x^n])^p\,dx\,\longrightarrow\,\frac{k}{e}\,\mathrm{Subst}\left[\int x^{k\,(m+1)-1}\left(a+b\,\mathrm{Sinh}\left[c+\frac{d\,x^{k\,n}}{e^n}\right]\right)^p dx,\,x,\,(e\,x)^{1/k}\right]$$

Program code:

```
Int[(e_.*x_)^m_*(a_.+b_.*Sinh[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  With[{k=Denominator[m]},
  k/e*Subst[Int[x^(k*(m+1)-1)*(a+b*Sinh[c+d*x^(k*n)/e^n])^p,x],x,(e*x)^(1/k)]] /;
FreeQ[{a,b,c,d,e},x] && IntegerQ[p] && IGtQ[n,0] && FractionQ[m]
```

```
Int[(e_.*x_)^m_*(a_.+b_.*Cosh[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  With[{k=Denominator[m]},
  k/e*Subst[Int[x^(k*(m+1)-1)*(a+b*Cosh[c+d*x^(k*n)/e^n])^p,x],x,(e*x)^(1/k)]] /;
FreeQ[{a,b,c,d,e},x] && IntegerQ[p] && IGtQ[n,0] && FractionQ[m]
```

6: $\int (e\,x)^m\,(a+b\,\text{Sinh}[c+d\,x^n])^p\,dx$ when $p-1\in\mathbb{Z}^+\wedge n\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $p-1\in\mathbb{Z}^+\wedge n\in\mathbb{Z}^+$, then

$$\int (e\,x)^m\,(a+b\,\text{Sinh}[c+d\,x^n])^p\,dx \longrightarrow \int (e\,x)^m\,\text{TrigReduce}\left[(a+b\,\text{Sinh}[c+d\,x^n])^p,\,x\right]dx$$

Program code:

```
Int[(e_.*x_)^m_.*(a_.+b_.*Sinh[c_.+d_.*x_^n_])^p_,x_Symbol] :=
  Int[ExpandTrigReduce[(e*x)^m,(a+b*Sinh[c+d*x^n])^p,x],x] /;
FreeQ[{a,b,c,d,e,m},x] && IGtQ[p,1] && IGtQ[n,0]
```

```
Int[(e_.*x_)^m_.*(a_.+b_.*Cosh[c_.+d_.*x_^n_])^p_,x_Symbol] :=
  Int[ExpandTrigReduce[(e*x)^m,(a+b*Cosh[c+d*x^n])^p,x],x] /;
FreeQ[{a,b,c,d,e,m},x] && IGtQ[p,1] && IGtQ[n,0]
```

3. $\int (e\,x)^m\,(a+b\,\text{Sinh}[c+d\,x^n])^p\,dx$ when $p<-1$

1: $\int x^m\,\text{Sinh}[a+b\,x^n]^p\,dx$ when $m-2\,n+1==0\,\land\,p<-1\,\land\,p\neq-2$

Reference: G&R 2.477.1 special case when $m-2\,n+1=0$

Reference: G&R 2.477.2' special case with $m-2\,n+1=0$

Rule: If $m-2\,n+1==0\,\land\,p<-1\,\land\,p\neq-2$, then

$$\int x^m\,\text{Sinh}[a+b\,x^n]^p\,dx\to\frac{x^n\,\text{Cosh}[a+b\,x^n]\,\text{Sinh}[a+b\,x^n]^{p+1}}{b\,n\,(p+1)}-\frac{n\,\text{Sinh}[a+b\,x^n]^{p+2}}{b^2\,n^2\,(p+1)\,(p+2)}-\frac{p+2}{p+1}\int x^m\,\text{Sinh}[a+b\,x^n]^{p+2}\,dx$$

Program code:

```
Int[x_^m_.*Sinh[a_.+b_.*x_^n_]^p_,x_Symbol] :=
  x^n*Cosh[a+b*x^n]*Sinh[a+b*x^n]^(p+1)/(b*n*(p+1)) -
  n*Sinh[a+b*x^n]^(p+2)/(b^2*n^2*(p+1)*(p+2)) -
  (p+2)/(p+1)*Int[x^m*Sinh[a+b*x^n]^(p+2),x] /;
FreeQ[{a,b,m,n},x] && EqQ[m-2*n+1,0] && LtQ[p,-1] && NeQ[p,-2]
```

```
Int[x_^m_.*Cosh[a_.+b_.*x_^n_]^p_,x_Symbol] :=
  -x^n*Sinh[a+b*x^n]*Cosh[a+b*x^n]^(p+1)/(b*n*(p+1)) +
  n*Cosh[a+b*x^n]^(p+2)/(b^2*n^2*(p+1)*(p+2)) +
  (p+2)/(p+1)*Int[x^m*Cosh[a+b*x^n]^(p+2),x] /;
FreeQ[{a,b,m,n},x] && EqQ[m-2*n+1,0] && LtQ[p,-1] && NeQ[p,-2]
```

2: $\int x^m \operatorname{Sinh}[a+b x^n]^p \, dx$ when $(m \mid n) \in \mathbb{Z} \land p<-1 \land p \neq -2 \land 0<2n<m+1$

Reference: G&R 2.477.1

Reference: G&R 2.477.2

Rule: If $(m \mid n) \in \mathbb{Z} \land p<-1 \land p \neq -2 \land 0<2n<m+1$, then

$$\int x^m \operatorname{Sinh}[a+b x^n]^p \, dx \longrightarrow$$

$$\frac{x^{m-n+1} \operatorname{Cosh}[a+b x^n] \operatorname{Sinh}[a+b x^n]^{p+1}}{b n (p+1)} - \frac{(m-n+1) x^{m-2n+1} \operatorname{Sinh}[a+b x^n]^{p+2}}{b^2 n^2 (p+1)(p+2)} - \frac{p+2}{p+1} \int x^m \operatorname{Sinh}[a+b x^n]^{p+2} \, dx + \frac{(m-n+1)(m-2n+1)}{b^2 n^2 (p+1)(p+2)} \int x^{m-2n} \operatorname{Sinh}[a+b x^n]^{p+2} \, dx$$

–

Program code:

```
Int[x_^m_.*Sinh[a_.+b_.*x_^n_]^p_,x_Symbol] :=
  x^(m-n+1)*Cosh[a+b*x^n]*Sinh[a+b*x^n]^(p+1)/(b*n*(p+1)) -
  (m-n+1)*x^(m-2*n+1)*Sinh[a+b*x^n]^(p+2)/(b^2*n^2*(p+1)*(p+2)) -
  (p+2)/(p+1)*Int[x^m*Sinh[a+b*x^n]^(p+2),x] +
  (m-n+1)*(m-2*n+1)/(b^2*n^2*(p+1)*(p+2))*Int[x^(m-2*n)*Sinh[a+b*x^n]^(p+2),x] /;
FreeQ[{a,b},x] && IntegersQ[m,n] && LtQ[p,-1] && NeQ[p,-2] && LtQ[0,2*n,m+1]
```

```
Int[x_^m_.*Cosh[a_.+b_.*x_^n_]^p_,x_Symbol] :=
  -x^(m-n+1)*Sinh[a+b*x^n]*Cosh[a+b*x^n]^(p+1)/(b*n*(p+1)) +
  (m-n+1)*x^(m-2*n+1)*Cosh[a+b*x^n]^(p+2)/(b^2*n^2*(p+1)*(p+2)) +
  (p+2)/(p+1)*Int[x^m*Cosh[a+b*x^n]^(p+2),x] -
  (m-n+1)*(m-2*n+1)/(b^2*n^2*(p+1)*(p+2))*Int[x^(m-2*n)*Cosh[a+b*x^n]^(p+2),x] /;
FreeQ[{a,b},x] && IntegersQ[m,n] && LtQ[p,-1] && NeQ[p,-2] && LtQ[0,2*n,m+1]
```

2. $\int (e\,x)^m\,(a+b\,\text{Sinh}[c+d\,x^n])^p\,dx$ when $p\in\mathbb{Z}\land n\in\mathbb{Z}^-$

   1. $\int (e\,x)^m\,(a+b\,\text{Sinh}[c+d\,x^n])^p\,dx$ when $p\in\mathbb{Z}\land n\in\mathbb{Z}^-\land m\in\mathbb{Q}$

      1: $\int x^m\,(a+b\,\text{Sinh}[c+d\,x^n])^p\,dx$ when $p\in\mathbb{Z}\land n\in\mathbb{Z}^-\land m\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: If $n\in\mathbb{Z}\land m\in\mathbb{Z}$, then $x^m\,F[x^n] = -\text{Subst}\left[\frac{F[x^{-n}]}{x^{m+2}},\,x,\,\frac{1}{x}\right]\partial_x\frac{1}{x}$

Rule: If $p\in\mathbb{Z}\land n\in\mathbb{Z}^-\land m\notin\mathbb{Q}$, then

$$\int x^m\,(a+b\,\text{Sinh}[c+d\,x^n])^p\,dx\;\longrightarrow\;-\text{Subst}\left[\int\frac{(a+b\,\text{Sinh}[c+d\,x^{-n}])^p}{x^{m+2}}\,dx,\,x,\,\frac{1}{x}\right]$$

Program code:

```
Int[x_^m_.*(a_.+b_.*Sinh[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  -Subst[Int[(a+b*Sinh[c+d*x^(-n)])^p/x^(m+2),x],x,1/x] /;
FreeQ[{a,b,c,d},x] && IntegerQ[p] && ILtQ[n,0] && IntegerQ[m]
```

```
Int[x_^m_.*(a_.+b_.*Cosh[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  -Subst[Int[(a+b*Cosh[c+d*x^(-n)])^p/x^(m+2),x],x,1/x] /;
FreeQ[{a,b,c,d},x] && IntegerQ[p] && ILtQ[n,0] && IntegerQ[m]
```

2: $\int (e\,x)^m \left(a+b\,\mathrm{Sinh}\left[c+d\,x^n\right]\right)^p \,dx$ when $p\in\mathbb{Z}\,\land\,n\in\mathbb{Z}^-\,\land\,m\in\mathbb{F}$

Derivation: Integration by substitution

Basis: If $n\in\mathbb{Z}\,\land\,k>1$, then $(e\,x)^m\,F[x^n] = -\frac{k}{e}\,\mathrm{Subst}\left[\frac{F\left[e^{-n}\,x^{-k\,n}\right]}{x^{k\,(m+1)+1}},\,x,\,\frac{1}{(e\,x)^{1/k}}\right]\partial_x\frac{1}{(e\,x)^{1/k}}$

Rule: If $p\in\mathbb{Z}\,\land\,n\in\mathbb{Z}^-\,\land\,m\in\mathbb{F}$, let $k=\mathrm{Denominator}[m]$, then

$$\int (e\,x)^m \left(a+b\,\mathrm{Sinh}\left[c+d\,x^n\right]\right)^p dx \longrightarrow -\frac{k}{e}\,\mathrm{Subst}\left[\int\frac{\left(a+b\,\mathrm{Sinh}\left[c+d\,e^{-n}\,x^{-k\,n}\right]\right)^p}{x^{k\,(m+1)+1}}\,dx,\,x,\,\frac{1}{(e\,x)^{1/k}}\right]$$

Program code:

```
Int[(e_.*x_)^m_*(a_.+b_.*Sinh[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  With[{k=Denominator[m]},
  -k/e*Subst[Int[(a+b*Sinh[c+d/(e^n*x^(k*n))])^p/x^(k*(m+1)+1),x],x,1/(e*x)^(1/k)]] /;
FreeQ[{a,b,c,d,e},x] && IntegerQ[p] && ILtQ[n,0] && FractionQ[m]
```

```
Int[(e_.*x_)^m_*(a_.+b_.*Cosh[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  With[{k=Denominator[m]},
  -k/e*Subst[Int[(a+b*Cosh[c+d/(e^n*x^(k*n))])^p/x^(k*(m+1)+1),x],x,1/(e*x)^(1/k)]] /;
FreeQ[{a,b,c,d,e},x] && IntegerQ[p] && ILtQ[n,0] && FractionQ[m]
```

2: $\int (e\,x)^m\,(a+b\,\mathrm{Sinh}[c+d\,x^n])^p\,dx$ when $p\in\mathbb{Z}\,\land\,n\in\mathbb{Z}^-\,\land\,m\notin\mathbb{Q}$

Derivation: Piecewise constant extraction and integration by substitution

Basis: $\partial_x\left((e\,x)^m\,\left(x^{-1}\right)^m\right)=0$

Basis: $F[x]=-\mathrm{Subst}\left[\frac{F\left[x^{-1}\right]}{x^2},\,x,\,\frac{1}{x}\right]\partial_x\frac{1}{x}$

Rule: If $p\in\mathbb{Z}\,\land\,n\in\mathbb{Z}^-\,\land\,m\notin\mathbb{Q}$, then

$$\int (e\,x)^m\,(a+b\,\mathrm{Sinh}[c+d\,x^n])^p\,dx\;\longrightarrow\;(e\,x)^m\,\left(x^{-1}\right)^m\int\frac{(a+b\,\mathrm{Sinh}[c+d\,x^n])^p}{\left(x^{-1}\right)^m}\,dx\;\longrightarrow\;-(e\,x)^m\,\left(x^{-1}\right)^m\,\mathrm{Subst}\left[\int\frac{(a+b\,\mathrm{Sinh}[c+d\,x^{-n}])^p}{x^{m+2}}\,dx,\,x,\,\frac{1}{x}\right]$$

Program code:

```
Int[(e_.*x_)^m_*(a_.+b_.*Sinh[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  -(e*x)^m*(x^(-1))^m*Subst[Int[(a+b*Sinh[c+d*x^(-n)])^p/x^(m+2),x],x,1/x] /;
FreeQ[{a,b,c,d,e,m},x] && IntegerQ[p] && ILtQ[n,0] && Not[RationalQ[m]]
```

```
Int[(e_.*x_)^m_*(a_.+b_.*Cosh[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  -(e*x)^m*(x^(-1))^m*Subst[Int[(a+b*Cosh[c+d*x^(-n)])^p/x^(m+2),x],x,1/x] /;
FreeQ[{a,b,c,d,e,m},x] && IntegerQ[p] && ILtQ[n,0] && Not[RationalQ[m]]
```

3. $\int (e\,x)^m\,(a+b\,\text{Sinh}[c+d\,x^n])^p\,dx$ when $p\in\mathbb{Z}\,\land\,n\in\mathbb{F}$

1: $\int x^m\,(a+b\,\text{Sinh}[c+d\,x^n])^p\,dx$ when $p\in\mathbb{Z}\,\land\,n\in\mathbb{F}$

Derivation: Integration by substitution

Basis: If $k\in\mathbb{Z}^+$, then $x^m\,F[x^n] == k\,\text{Subst}\left[x^{k\,(m+1)-1}\,F\left[x^{k\,n}\right],\,x,\,x^{1/k}\right]\,\partial_x x^{1/k}$

Rule: If $p\in\mathbb{Z}\,\land\,n\in\mathbb{F}$, let $k=\text{Denominator}[n]$, then

$$\int x^m\,(a+b\,\text{Sinh}[c+d\,x^n])^p\,dx\,\longrightarrow\,k\,\text{Subst}\left[\int x^{k\,(m+1)-1}\,\left(a+b\,\text{Sinh}\left[c+d\,x^{k\,n}\right]\right)^p\,dx,\,x,\,x^{1/k}\right]$$

Program code:

```
Int[x_^m_.*(a_.+b_.*Sinh[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  Module[{k=Denominator[n]},
  k*Subst[Int[x^(k*(m+1)-1)*(a+b*Sinh[c+d*x^(k*n)])^p,x],x,x^(1/k)]] /;
FreeQ[{a,b,c,d,m},x] && IntegerQ[p] && FractionQ[n]
```

```
Int[x_^m_.*(a_.+b_.*Cosh[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  Module[{k=Denominator[n]},
  k*Subst[Int[x^(k*(m+1)-1)*(a+b*Cosh[c+d*x^(k*n)])^p,x],x,x^(1/k)]] /;
FreeQ[{a,b,c,d,m},x] && IntegerQ[p] && FractionQ[n]
```

2: $\int (e\,x)^m\,(a+b\,\text{Sinh}[c+d\,x^n])^p\,dx$ when $p\in\mathbb{Z}\,\land\,n\in\mathbb{F}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(e\,x)^m}{x^m} == 0$

Rule: If $p\in\mathbb{Z}\,\land\,n\in\mathbb{F}$, then

$$\int (e\,x)^m \left(a+b\,\text{Sinh}\left[c+d\,x^n\right]\right)^p dx \,\to\, \frac{e^{\text{IntPart}[m]}\,(e\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}} \int x^m \left(a+b\,\text{Sinh}\left[c+d\,x^n\right]\right)^p dx$$

– Program code:

```
Int[(e_*x_)^m_*(a_.+b_.*Sinh[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  e^IntPart[m]*(e*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a+b*Sinh[c+d*x^n])^p,x] /;
FreeQ[{a,b,c,d,e,m},x] && IntegerQ[p] && FractionQ[n]
```

```
Int[(e_*x_)^m_*(a_.+b_.*Cosh[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  e^IntPart[m]*(e*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a+b*Cosh[c+d*x^n])^p,x] /;
FreeQ[{a,b,c,d,e,m},x] && IntegerQ[p] && FractionQ[n]
```

4. $\int (e\,x)^m \left(a+b\,\text{Sinh}\left[c+d\,x^n\right]\right)^p dx$ when $p\in\mathbb{Z} \land m\neq -1 \land \frac{n}{m+1}\in\mathbb{Z}^+$

1: $\int x^m \left(a+b\,\text{Sinh}\left[c+d\,x^n\right]\right)^p dx$ when $p\in\mathbb{Z} \land m\neq -1 \land \frac{n}{m+1}\in\mathbb{Z}^+$

Derivation: Integration by substitution

Basis: If $\frac{n}{m+1}\in\mathbb{Z}$, then $x^m\,F\left[x^n\right] = \frac{1}{m+1}\,\text{Subst}\left[F\left[x^{\frac{n}{m+1}}\right],\,x,\,x^{m+1}\right]\partial_x x^{m+1}$

– Rule: If $p\in\mathbb{Z} \land m\neq -1 \land \frac{n}{m+1}\in\mathbb{Z}^+$, then

$$\int x^m \left(a+b\,\text{Sinh}\left[c+d\,x^n\right]\right)^p dx \,\to\, \frac{1}{m+1}\,\text{Subst}\left[\int\left(a+b\,\text{Sinh}\left[c+d\,x^{\frac{n}{m+1}}\right]\right)^p dx,\,x,\,x^{m+1}\right]$$

– Program code:

```
Int[x_^m_.*(a_.+b_.*Sinh[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  1/(m+1)*Subst[Int[(a+b*Sinh[c+d*x^Simplify[n/(m+1)]])^p,x],x,x^(m+1)] /;
FreeQ[{a,b,c,d,m,n},x] && IntegerQ[p] && NeQ[m,-1] && IGtQ[Simplify[n/(m+1)],0] && Not[IntegerQ[n]]
```

```
Int[x_^m_.*(a_.+b_.*Cosh[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  1/(m+1)*Subst[Int[(a+b*Cosh[c+d*x^Simplify[n/(m+1)]])^p,x],x,x^(m+1)] /;
FreeQ[{a,b,c,d,m,n},x] && IntegerQ[p] && NeQ[m,-1] && IGtQ[Simplify[n/(m+1)],0] && Not[IntegerQ[n]]
```

2: $\int (e\,x)^m\,(a+b\,\mathrm{Sinh}[c+d\,x^n])^p\,dx$ when $p\in\mathbb{Z}\,\wedge\,m\neq-1\,\wedge\,\frac{n}{m+1}\in\mathbb{Z}^+$

Derivation: Piecewise constant extraction

Basis: $\partial_x\frac{(e\,x)^m}{x^m}=0$

Rule: If $p\in\mathbb{Z}\,\wedge\,m\neq-1\,\wedge\,\frac{n}{m+1}\in\mathbb{Z}^+$, then

$$\int (e\,x)^m\,(a+b\,\mathrm{Sinh}[c+d\,x^n])^p\,dx\,\longrightarrow\,\frac{e^{\mathrm{IntPart}[m]}\,(e\,x)^{\mathrm{FracPart}[m]}}{x^{\mathrm{FracPart}[m]}}\int x^m\,(a+b\,\mathrm{Sinh}[c+d\,x^n])^p\,dx$$

Program code:

```
Int[(e_*x_)^m_*(a_.+b_.*Sinh[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  e^IntPart[m]*(e*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a+b*Sinh[c+d*x^n])^p,x] /;
FreeQ[{a,b,c,d,e,m,n},x] && IntegerQ[p] && NeQ[m,-1] && IGtQ[Simplify[n/(m+1)],0] && Not[IntegerQ[n]]
```

```
Int[(e_*x_)^m_*(a_.+b_.*Cosh[c_.+d_.*x_^n_])^p_.,x_Symbol] :=
  e^IntPart[m]*(e*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a+b*Cosh[c+d*x^n])^p,x] /;
FreeQ[{a,b,c,d,e,m,n},x] && IntegerQ[p] && NeQ[m,-1] && IGtQ[Simplify[n/(m+1)],0] && Not[IntegerQ[n]]
```

5. $\int (e\,x)^m \left(a + b\,\text{Sinh}\left[c + d\,x^n\right]\right)^p\,dx$ when $p \in \mathbb{Z}^+$

1: $\int (e\,x)^m\,\text{Sinh}\left[c + d\,x^n\right]\,dx$

Derivation: Algebraic expansion

Basis: $\text{Sinh}[z] == \frac{e^z}{2} - \frac{e^{-z}}{2}$

Basis: $\text{Cosh}[z] == \frac{e^z}{2} + \frac{e^{-z}}{2}$

Rule:

$$\int (e\,x)^m\,\text{Sinh}\left[c + d\,x^n\right]\,dx \longrightarrow \frac{1}{2}\int (e\,x)^m\,e^{c+d\,x^n}\,dx - \frac{1}{2}\int (e\,x)^m\,e^{-c-d\,x^n}\,dx$$

Program code:

```
Int[(e_.*x_)^m_.*Sinh[c_.+d_.*x_^n_],x_Symbol] :=
  1/2*Int[(e*x)^m*E^(c+d*x^n),x] - 1/2*Int[(e*x)^m*E^(-c-d*x^n),x] /;
FreeQ[{c,d,e,m,n},x]
```

```
Int[(e_.*x_)^m_.*Cosh[c_.+d_.*x_^n_],x_Symbol] :=
  1/2*Int[(e*x)^m*E^(c+d*x^n),x] + 1/2*Int[(e*x)^m*E^(-c-d*x^n),x] /;
FreeQ[{c,d,e,m,n},x]
```

2: $\int (e\,x)^m \left(a+b\,\text{Sinh}\left[c+d\,x^n\right]\right)^p dx$ when $p \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $p \in \mathbb{Z}^+$, then

$$\int (e\,x)^m \left(a+b\,\text{Sinh}\left[c+d\,x^n\right]\right)^p dx \longrightarrow \int (e\,x)^m\,\text{TrigReduce}\left[\left(a+b\,\text{Sinh}\left[c+d\,x^n\right]\right)^p,\,x\right] dx$$

Program code:

```
Int[(e_.*x_)^m_.*(a_.+b_.*Sinh[c_.+d_.*x_^n_])^p_,x_Symbol] :=
  Int[ExpandTrigReduce[(e*x)^m,(a+b*Sinh[c+d*x^n])^p,x],x] /;
FreeQ[{a,b,c,d,e,m,n},x] && IGtQ[p,0]
```

```
Int[(e_.*x_)^m_.*(a_.+b_.*Cosh[c_.+d_.*x_^n_])^p_,x_Symbol] :=
  Int[ExpandTrigReduce[(e*x)^m,(a+b*Cosh[c+d*x^n])^p,x],x] /;
FreeQ[{a,b,c,d,e,m,n},x] && IGtQ[p,0]
```

S: $\int x^m \left(a + b \operatorname{Sinh}\left[c + d u^n\right]\right)^p dx$ when $u == f + g x \land m \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: If $m \in \mathbb{Z}$, then $x^m F[f + g x] == \frac{1}{g^{m+1}} \operatorname{Subst}[(x - f)^m F[x], x, f + g x] \partial_x (f + g x)$

Rule: If $u == f + g x \land m \in \mathbb{Z}$, then

$$\int x^m \left(a + b \operatorname{Sinh}\left[c + d u^n\right]\right)^p dx \longrightarrow \frac{1}{g^{m+1}} \operatorname{Subst}\left[\int (x - f)^m \left(a + b \operatorname{Sinh}\left[c + d x^n\right]\right)^p dx, x, u\right]$$

Program code:

```
Int[x_^m_.*(a_.+b_.*Sinh[c_.+d_.*u_^n_])^p_.,x_Symbol] :=
  1/Coefficient[u,x,1]^(m+1)*Subst[Int[(x-Coefficient[u,x,0])^m*(a+b*Sinh[c+d*x^n])^p,x],x,u] /;
FreeQ[{a,b,c,d,n,p},x] && LinearQ[u,x] && NeQ[u,x] && IntegerQ[m]
```

```
Int[x_^m_.*(a_.+b_.*Cosh[c_.+d_.*u_^n_])^p_.,x_Symbol] :=
  1/Coefficient[u,x,1]^(m+1)*Subst[Int[(x-Coefficient[u,x,0])^m*(a+b*Cosh[c+d*x^n])^p,x],x,u] /;
FreeQ[{a,b,c,d,n,p},x] && LinearQ[u,x] && NeQ[u,x] && IntegerQ[m]
```

X: $\int (e\,x)^m\,(a+b\,\mathrm{Sinh}[c+d\,u^n])^p\,dx$ when $u == f+g\,x$

Rule:

$$\int (e\,x)^m\,(a+b\,\mathrm{Sinh}[c+d\,u^n])^p\,dx \longrightarrow \int (e\,x)^m\,(a+b\,\mathrm{Sinh}[c+d\,u^n])^p\,dx$$

Program code:

```
Int[(e_.*x_)^m_.*(a_.+b_.*Sinh[c_.+d_.*u_^n_])^p_.,x_Symbol] :=
  Unintegrable[(e*x)^m*(a+b*Sinh[c+d*u^n])^p,x] /;
FreeQ[{a,b,c,d,e,m,n,p},x] && LinearQ[u,x]
```

```
Int[(e_.*x_)^m_.*(a_.+b_.*Cosh[c_.+d_.*u_^n_])^p_.,x_Symbol] :=
  Unintegrable[(e*x)^m*(a+b*Cosh[c+d*u^n])^p,x] /;
FreeQ[{a,b,c,d,e,m,n,p},x] && LinearQ[u,x]
```

N: $\int (e\,x)^m\,(a+b\,\mathrm{Sinh}[u])^p\,dx$ when $u == c+d\,x^n$

Derivation: Algebraic normalization

Rule: If $u == c+d\,x^n$, then

$$\int (e\,x)^m\,(a+b\,\mathrm{Sinh}[u])^p\,dx \longrightarrow \int (e\,x)^m\,(a+b\,\mathrm{Sinh}[c+d\,x^n])^p\,dx$$

Program code:

```
Int[(e_*x_)^m_.*(a_.+b_.*Sinh[u_])^p_.,x_Symbol] :=
  Int[(e*x)^m*(a+b*Sinh[ExpandToSum[u,x]])^p,x] /;
FreeQ[{a,b,e,m,p},x] && BinomialQ[u,x] && Not[BinomialMatchQ[u,x]]
```

```
Int[(e_*x_)^m_.*(a_.+b_.*Cosh[u_])^p_.,x_Symbol] :=
  Int[(e*x)^m*(a+b*Cosh[ExpandToSum[u,x]])^p,x] /;
FreeQ[{a,b,e,m,p},x] && BinomialQ[u,x] && Not[BinomialMatchQ[u,x]]
```

Rules for integrands of the form $x^m \operatorname{Sinh}[a+b x^n]^p \operatorname{Cosh}[a+b x^n]$

1: $\int x^{n-1} \operatorname{Sinh}[a+b x^n]^p \operatorname{Cosh}[a+b x^n]\, dx$ when $p \neq -1$

Derivation: Power rule for integration

Rule: If $p \neq -1$, then

$$\int x^{n-1} \operatorname{Sinh}[a+b x^n]^p \operatorname{Cosh}[a+b x^n]\, dx \longrightarrow \frac{\operatorname{Sinh}[a+b x^n]^{p+1}}{b\, n\, (p+1)}$$

Program code:

```
Int[x_^m_.*Sinh[a_.+b_.*x_^n_]^p_.*Cosh[a_.+b_.*x_^n_.],x_Symbol] :=
  Sinh[a+b*x^n]^(p+1)/(b*n*(p+1)) /;
FreeQ[{a,b,m,n,p},x] && EqQ[m,n-1] && NeQ[p,-1]
```

```
Int[x_^m_.*Cosh[a_.+b_.*x_^n_]^p_.*Sinh[a_.+b_.*x_^n_.],x_Symbol] :=
  Cosh[a+b*x^n]^(p+1)/(b*n*(p+1)) /;
FreeQ[{a,b,m,n,p},x] && EqQ[m,n-1] && NeQ[p,-1]
```

2: $\int x^m \operatorname{Sinh}[a+b x^n]^p \operatorname{Cosh}[a+b x^n]\, dx$ when $0<n<m+1 \land p \neq -1$

Reference: G&R 2.479.6

Reference: G&R 2.479.3

Derivation: Integration by parts

Basis: $x^m \operatorname{Cosh}[a+b x^n] \operatorname{Sinh}[a+b x^n]^p == x^{m-n+1} \partial_x \frac{\operatorname{Sinh}[a+b x^n]^{p+1}}{b\, n\, (p+1)}$

Rule: If $0<n<m+1 \land p \neq -1$, then

$$\int x^m \operatorname{Sinh}[a+b x^n]^p \operatorname{Cosh}[a+b x^n]\, dx \to \frac{x^{m-n+1} \operatorname{Sinh}[a+b x^n]^{p+1}}{b\, n\, (p+1)} - \frac{m-n+1}{b\, n\, (p+1)} \int x^{m-n} \operatorname{Sinh}[a+b x^n]^{p+1}\, dx$$

Program code:

```
Int[x_^m_.*Sinh[a_.+b_.*x_^n_.]^p_.*Cosh[a_.+b_.*x_^n_.],x_Symbol] :=
  x^(m-n+1)*Sinh[a+b*x^n]^(p+1)/(b*n*(p+1)) -
  (m-n+1)/(b*n*(p+1))*Int[x^(m-n)*Sinh[a+b*x^n]^(p+1),x] /;
FreeQ[{a,b,p},x] && LtQ[0,n,m+1] && NeQ[p,-1]
```

```
Int[x_^m_.*Cosh[a_.+b_.*x_^n_.]^p_.*Sinh[a_.+b_.*x_^n_.],x_Symbol] :=
  x^(m-n+1)*Cosh[a+b*x^n]^(p+1)/(b*n*(p+1)) -
  (m-n+1)/(b*n*(p+1))*Int[x^(m-n)*Cosh[a+b*x^n]^(p+1),x] /;
FreeQ[{a,b,p},x] && LtQ[0,n,m+1] && NeQ[p,-1]
```